Using computers: spreadsheets to teach terminal velocity

Many high school students in the UK are required to understand terminal velocity in the contexts of a falling object (the classic example being a parachutist) and top speeds for vehicles. The forces acting on a parachutist can be represented as in figure 1.

Terminal velocity is traditionally a difficult topic for students to understand. At least part of the difficulty arises because the drag force is velocity-dependent. In other words, there is a feedback loop, as follows:

"an increase in velocity causes
an increase in the drag force, which causes
a decrease in the downwards resultant force, which causes
a decrease in acceleration, which causes
a smaller increase in velocity, which causes..."

By modelling this feedback on a spreadsheet, students' understanding of terminal velocity can be enhanced. Figure 2 shows an example of a spreadsheet the students can construct, with guidance from an appropriate worksheet and after suitable class teaching. For some students, an introductory exercise on entering formulae into spreadsheets is necessary.

Figure 3 shows the first few lines of the formulae entered into the spreadsheet in order to generate the numbers in figure 2.

The time column 'counts' in intervals of 1 s, by adding 1 onto the previous value each time. The weight is calculated as $W = mg$ and the acceleration is calculated from $F = ma$, remembering that F is the resultant force and is thus the difference between the weight and the drag.



Figure 1. *Representation of the forces on a parachutist.*

It is important to note two things at this stage.

- each velocity is calculated by adding the acceleration to the previous value of the velocity. This is applicable (although not perfectly accurate) because the time intervals are 1 second and the acceleration is the change of velocity per second. For example, if the previous velocity is 10 m s^{-1}, and the acceleration is 8 m s^{-2} for the next second, then the next velocity will be 18 m s^{-1}.
- the drag force is dependent on the velocity (the v^2 dependence need not be understood by students; it is sufficient that they appreciate that 'as v goes up, so does the drag force...'). The

	A	B	C	D	E	F	G	H	I
1			Terminal velocity						
2									
3									
4	mass (kg)		time	velocity	weight	drag	resultant	acceleration	
5	50		(s)	(m/s)	(N)	(N)	(N)	(m/s2)	
6	coeff		0	0.00	500	0.0	500.0	10.000	
7	0.2		1	10.00	500	20.0	480.0	9.600	
8			2	19.60	500	76.8	423.2	8.463	
9			3	28.06	500	157.5	342.5	6.850	
10			4	34.91	500	243.8	256.2	5.124	
11			5	40.04	500	320.6	179.4	3.588	
12			6	43.63	500	380.6	119.4	2.387	
13			7	46.01	500	423.4	76.6	1.531	
14			8	47.54	500	452.1	47.9	0.958	
15			9	48.50	500	470.5	29.5	0.590	
16			10	49.09	500	482.0	18.0	0.360	
17			11	49.45	500	489.1	10.9	0.218	
18			12	49.67	500	493.4	6.6	0.132	
19			13	49.80	500	496.0	4.0	0.079	
20			14	49.88	500	497.6	2.4	0.048	
21			15	49.93	500	498.6	1.4	0.029	

Figure 2. *An example of a spreadsheet to model terminal velocity.*

'drag coefficient' depends on the viscosity of the air and the surface area of the parachutist. In this model, 0.2 is a value that is found empirically to provide good results.

These two points are what causes the feedback in the situation.

Figure 4 shows the resulting velocity–time graph generated from the spreadsheet of figure 1. This can be done very easily in Excel using the 'chart wizard' and following the step-by step instructions. Note the decrease of the gradient with time. The speed does indeed level out at a 'terminal velocity', in this case around 50 m s^{-1}.

We can see that physical parameters can easily be changed and investigated. In the terminal velocity model, the mass of the object, or the drag coefficient, can be changed and the results can be seen instantly. A neat way to do this on the spreadsheet is to define a name for a cell and enter the value of that quantity. This has been done in the top-left hand corner of each of the spreadsheets in this article. To do this use INSERT–NAME–DEFINE. Then use that name in the formulae you enter. Changing the value of this parameter in the defined square will then change it wherever it appears in a formula. You do not then need to change its value in several places. For example, varying the mass requires changing both columns E and H in figures 2 and 3. Using the 'define name' function and altering the value of cell A5 achieves this in one step, rather than two.

Once a basic velocity–time graph has been obtained the activity can be extended in many ways. Some of these are as follows:

- Plot an acceleration–time graph on the same axes as the velocity–time graph. Students will then be able to see easily that a reduction in the gradient of the velocity–time graph is linked to decreasing acceleration. In particular, maximum velocity is achieved when the acceleration falls to zero.
- Investigate the effect of changing the mass. It will be seen that the greater the mass, the greater the terminal velocity. Students can be asked to look at the data in the spreadsheet to try to work out why this might be.
- Plot a graph of the motion that would occur on the Moon. To do this the value of g needs to be changed, and the drag coefficient should be set at zero (assuming there is no atmosphere on the Moon). The resulting lack of a velocity-

	A	B	C	D	E	F	G	H
1			Terminal velocity					
2								
3								
4	mass (kg)		time	velocity	weight	drag	resultant	acceleration
5	50		(s)	(m/s)	(N)	(N)	(N)	(m/s2)
6	coeff		0	0	=mass*10	=coeff*(D6^2)	=E6-F6	=G6/mass
7	0.2		=C6+1	=D6+H6	=mass*10	=coeff*(D7^2)	=E7-F7	=G7/mass
8			=C7+1	=D7+H7	=mass*10	=coeff*(D8^2)	=E8-F8	=G8/mass
9			=C8+1	=D8+H8	=mass*10	=coeff*(D9^2)	=E9-F9	=G9/mass
10			=C9+1	=D9+H9	=mass*10	=coeff*(D10^2)	=E10-F10	=G10/mass

Figure 3. *The first few lines of the formulae to generate the numbers in figure 2.*



Figure 4. *Speed–time graph for a falling body with a v^2 drag force.*

dependent drag force takes away the feedback in the situation, and uniformly accelerated motion is the result. This is very clear from looking at the data in the spreadsheet. It is more interesting if you can plot the graph on the same axes as the graph for the Earth.

- Investigate what happens when the parachute is opened. To do this increase the value of the drag coefficient halfway down the spreadsheet, at, for example, $t = 10$ seconds. Trial-and-error is required here to get a value of the drag coefficient that gives good results, but it should be possible to generate a second, lower, terminal velocity.

I have found that this technique has worked with a wide variety of abilities in Year 10 classes in the UK. It seems to enhance students' understanding of a difficult topic, and the number of possible extension tasks means that it can be used to challenge the most able.

Michael Lingard
Grenville College, Bideford, Devon, UK